Assignment 4 Question 3   Determinants

```
> restart;
  P := () -> randpoly(x,degree=2,dense):
> A := Matrix(3,3,P);
```

$$A := \begin{bmatrix} -7x^2+22x-55 & -94x^2+87x-56 & 97-62x \\ -73x^2-4x-83 & -10x^2+62x-82 & 80x^2-44x+71 \\ -17x^2-75x-10 & -7x^2-40x+42 & -50x^2+23x+75 \end{bmatrix}$$

By inspection we can see that deg(det(A)) is at most 6. Assuming the biggest coefficient in the matrix is m digits base B and the biggest degree of a polynomial is d and the matrix is n by n, we showed in class that a bound on the height of the determinant of A is given by

```
> B^(n*m)*(d+1)^(n-1)*n!;
```

$$B^{nm}(d+1)^{n-1}n!$$

Substituting 94 for B^m and 2 for d and 3 for n we have

```
> bound := 94^3*3^2*3!;
```

$$bound := 44851536$$

```
> Primes := []:
  M := 1:
  p := 101:
  while M < 2*bound do
     Primes := [op(Primes),p]; M := M*p; p := nextprime(p);
  od:
  Primes;
  M;
```

$$[101, 103, 107, 109]$$

$$121330189$$

```
> phix := proc(a,alpha) Eval(a,x=alpha) mod p end;
```

$$phix := \mathbf{proc}(a, \text{alpha})\ Eval(a, x=\text{alpha})\ \mathbf{mod}\ p\ \mathbf{end\ proc}$$

```
> for p in Primes do
      Ap := map( modp, A, p );
      for i from 0 to 6 do
          Api := map( phix, Ap, i );
          dp[i] := Det(Api) mod p;
      od;
      d[p] := Interp( [seq(i,i=0..6)], [seq(dp[i],i=0..6)], x ) mod p;
  od;
  d := chrem( [seq(d[p],p=Primes)], Primes );
```

$$Ap := \begin{bmatrix} 94x^2+22x+46 & 7x^2+87x+45 & 97+39x \\ 28x^2+97x+18 & 91x^2+62x+19 & 80x^2+57x+71 \\ 84x^2+26x+91 & 94x^2+61x+42 & 51x^2+23x+75 \end{bmatrix}$$

$$d_{101} := 31x^6 + 89x^5 + 62x^4 + 31x^3 + 24x^2 + 57x + 59$$

$$Ap := \begin{bmatrix} 96x^2 + 22x + 48 & 9x^2 + 87x + 47 & 97 + 41x \\ 30x^2 + 99x + 20 & 93x^2 + 62x + 21 & 80x^2 + 59x + 71 \\ 86x^2 + 28x + 93 & 96x^2 + 63x + 42 & 53x^2 + 23x + 75 \end{bmatrix}$$

$$d_{103} := 20x^6 + 50x^5 + 44x^4 + x^3 + 83x^2 + 98x + 72$$

$$Ap := \begin{bmatrix} 100x^2 + 22x + 52 & 13x^2 + 87x + 51 & 97 + 45x \\ 34x^2 + 103x + 24 & 97x^2 + 62x + 25 & 80x^2 + 63x + 71 \\ 90x^2 + 32x + 97 & 100x^2 + 67x + 42 & 57x^2 + 23x + 75 \end{bmatrix}$$

$$d_{107} := 103x^6 + 6x^5 + 44x^4 + 68x^3 + 13x^2 + 98x + 10$$

$$Ap := \begin{bmatrix} 102x^2 + 22x + 54 & 15x^2 + 87x + 53 & 97 + 47x \\ 36x^2 + 105x + 26 & 99x^2 + 62x + 27 & 80x^2 + 65x + 71 \\ 92x^2 + 34x + 99 & 102x^2 + 69x + 42 & 59x^2 + 23x + 75 \end{bmatrix}$$

$$d_{109} := 52x^6 + 9x^5 + x^4 + 89x^3 + 25x^2 + 49x + 60$$

$$d := 463520x^6 + 121254225x^5 + 120790204x^4 + 937816x^3 + 120874703x^2 + 55203x + 121105927$$

We have used the positive range in chrem. We will convert explicitly to the symmetric range.

```
> d := mods(d,M);
```

$$d := 463520x^6 - 75964x^5 - 539985x^4 + 937816x^3 - 455486x^2 + 55203x - 224262$$

```
> LinearAlgebra[Determinant](A);
```

$$463520x^6 - 75964x^5 - 539985x^4 + 937816x^3 - 455486x^2 + 55203x - 224262$$